```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the approximation of the steady two dimensional rotational isentropic flow studied in page 436 of R. Courant, D. Hilbert, *Methods of Mathematical Physics*, Wiley Classics Library, Wiley, 1989,

$$\left\{\begin{array}{l} u\,\rho \dfrac{\partial\,\omega}{\partial x} + c^2\,\dfrac{\partial\,\sigma}{\partial x} = 0, \\ u\,\rho \dfrac{\partial\,\lambda}{\partial x} + c^2\,\dfrac{\partial\,\sigma}{\partial y} = 0, \\ \rho\,\dfrac{\partial\,\omega}{\partial x} + \rho\,\dfrac{\partial\,\lambda}{\partial y} + u\,\dfrac{\partial \sigma}{\partial x} = 0, \end{array}\right. \tag{1}$$

where u denotes the constant velocity parallel to the x-axis, ρ the constant density and c the speed of sound. Let us introduce the ring $A = \mathbb{Q}(u, \rho, c)[d_x, d_y]$ of differential operators in d_x and d_y with coefficients in the field $\mathbb{Q}(u, \rho, c)$

```
> A:=DefineOreAlgebra(diff=[d[x],x],diff=[d[y],y],polynom=[x,y],
> comm=[u,rho,c]):
```

and the system matrix $R \in A^{3\times 3}$ of (1) defined by:

```
> R:=matrix(3,3,[[u*rho*d[x],c^2*d[x],0],[0,c^2*d[y],u*rho*d[x]],
> [rho*d[x],u*d[x],rho*d[y]]]);
```

$$R := \begin{bmatrix} u\,\rho\,d_x & c^2 d_x & 0 \\ 0 & c^2 d_y & u\,\rho\,d_x \\ \rho\,d_x & u\,d_x & \rho\,d_y \end{bmatrix}$$

We denote by $M = A^{1\times 3}/(A^{1\times 3}\,R)$ the A-module finitely presented by the matrix R. Let us study the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A): Endo[1];
```

$$\left[\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & c^2 & 0 \\ 0 & -u\,\rho & 0 \\ 0 & 0 & -u\,\rho \end{bmatrix}, \begin{bmatrix} 0 & 0 & c^2 u\,\rho \\ 0 & 0 & -u^2\rho^2 \\ 0 & -c^2\left(-c^2+u^2\right) & 0 \end{bmatrix}\right]$$

Hence, we obtain that E is finitely generated by three endomorphisms $f_1 = \mathrm{id}_M$, f_2 and f_3 defined by $f_i(\pi(\lambda)) = \pi(\lambda\,P_i)$, where $\pi : A^{1\times 3} \longrightarrow M$ denotes the projection onto M, $\lambda \in A^{1\times 3}$ and P_i is one of the three previous matrices. The generators f_i's of E satisfy the relations $Endo[2]\,(f_1 \quad f_2 \quad f_3)^T = 0$, where $Endo[2]$ is the following matrix:

```
> Endo[2];
```

$$\begin{bmatrix} -uc^2\,\rho\,d_x + u^3\,d_x\,\rho & 0 & -d_y \\ 0 & c^2 d_y & d_x \\ 0 & -c^2\,d_x + u^2\,d_x & d_y \end{bmatrix}$$

Let us study the idempotents of the endomorphism ring E defined by means of constant matrices, i.e., matrices defined by with zero-order differential operators:

```
> Idem:=IdempotentsConstCoeff(R,Endo[1],A,0,alpha);
```

$$Idem := [[\begin{bmatrix} 0 & -1/2\,\frac{c^2}{u\,\rho} & \alpha_1 c \\ 0 & 1/2 & -\frac{u\,\rho\,\alpha_1}{c} \\ 0 & -\frac{\alpha_1\,c\,(-c^2+u^2)}{u\,\rho} & 1/2 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & \frac{c^2}{u\,\rho} & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix},$$

$$\begin{bmatrix} 0 & -\frac{c^2}{u\,\rho} & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 1/2\,\frac{c^2}{u\,\rho} & \alpha_1 c \\ 0 & 1/2 & -\frac{u\,\rho\,\alpha_1}{c} \\ 0 & -\frac{\alpha_1\,c\,(-c^2+u^2)}{u\,\rho} & 1/2 \end{bmatrix}]$$

$$[Ore\_algebra, [\text{“}diff\text{''}, \text{“}diff\text{''}], [x,y], [d_x, d_y], [x,y], [u,\rho,c,\alpha_1], 0, [],$$
$$[-1-4\,\alpha_1^2\,c^2+4\,\alpha_1^2\,u^2], [x,y], [], [], [diff=[d_x,x], diff=[d_y,y]]]]$$

We obtain the two trivial idempotents 0 and id_M of E respectively defined by the matrices 0 or I_3, two non-trivial idempotents respectively defined by the matrices $Idem[1,3]$ and $Idem[1,4]$ whose entries belong to A and two non-trivial idempotents of $\mathrm{end}_B(B \otimes_A M)$, where $B = \mathbb{Q}(u,\rho,c)[\alpha_1]/(4\,(u^2-c^2)\,\alpha_1^2-1)[d_x,d_y]$, respectively defined by the matrices $Idem[1,1]$ and $Idem[1,6]$. Let us consider the matrix $P_1 = Idem[1,3]$ and Q_1 satisfying $R\,P_1 = Q_1\,R$:

```
> P[1]:=Idem[1,3]; Q[1]:=Factorize(Mult(R,P,A),R,A);
```

$$P_1 := \begin{bmatrix} 1 & \frac{c^2}{u\,\rho} & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad Q_1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 0 & 0 \\ u^{-1} & 0 & 0 \end{bmatrix}$$

We can check that we have $P_1^2 = P_1$ and $Q_1^2 = Q_1$:

```
> VERIF1:=simplify(evalm(Mult(P[1],P[1],A)-P[1]));
> VERIF2:=simplify(evalm(Mult(Q[1],Q[1],A)-Q[1]));
```

$$VERIF1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Using the fact that matrices P_1 and Q_1 are idempotents of $A^{1\times 3}$, we obtain that the A-modules $\ker_A(.P_1)$, $\mathrm{im}_A(.P_1) = \ker_A(.(I_3-P_1))$, $\ker_A(.Q_1)$ and $\mathrm{im}_A(.Q_1) = \ker_A(.(I_3-Q_1))$ are projective, and thus, free by the Quillen-Suslin theorem. As the entries of P_1 and Q_1 only belong to the field $\mathbb{Q}(u,\rho,c)$, using linear algebraic techniques, we can easily compute bases of the corresponding $\mathbb{Q}(u,\rho,c)$-vector spaces, and thus, bases over the ring A:

```
> U1:=SyzygyModule(P[1],A): U2:=SyzygyModule(evalm(1-P[1]),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[1],A): V2:=SyzygyModule(evalm(1-Q[1]),A):
> V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ u\,\rho & c^2 & 0 \end{bmatrix} \quad V := \begin{bmatrix} 1 & 0 & -u \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

The matrices $U \in \mathrm{GL}_3(A)$ and $V \in \mathrm{GL}_3(A)$ are such that the matrices $U\,P_1\,U^{-1}$ and $V\,Q_1\,V^{-1}$ are two block-diagonal matrices formed by the diagonal matrices 0_2 and 1:

```
> VERIF1:=Mult(U,P[1],LeftInverse(U,A),A);
> VERIF2:=Mult(V,Q[1],LeftInverse(V,A),A);
```

$$VERIF1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Then, the matrix R is equivalent to the block-diagonal matrix $V\,R\,U^{-1}$ defined by:

```
> R_dec:=Mult(V,R,LeftInverse(U,A),A);
```

$$R\_dec := \begin{bmatrix} -d_x\,(-c^2+u^2) & -u\,\rho\,d_y & 0 \\ c^2\,d_y & u\,\rho\,d_x & 0 \\ 0 & 0 & d_x \end{bmatrix}$$

This last result can be obtained using the command *HeuristicDecomposition*:

```
> HeuristicDecomposition(R,P[1],A)[1];
```

$$\begin{bmatrix} -d_x\,(-c^2+u^2) & -u\,\rho\,d_y & 0 \\ c^2\,d_y & u\,\rho\,d_x & 0 \\ 0 & 0 & d_x \end{bmatrix}$$

Let us now consider the first 2×2 block-diagonal matrix S of R_dec defined by:

```
> S:=submatrix(R_dec,1..2,1..2);
```

$$S := \begin{bmatrix} -d_x\,(-c^2+u^2) & -u\,\rho\,d_y \\ c^2\,d_y & u\,\rho\,d_x \end{bmatrix}$$

Let us try check whether or not the matrix S is equivalent to a block-diagonal matrix. To do that, we introduce the A-module $L = A^{1\times 2}/(A^{1\times 2}\,S)$ finitely presented by the matrix S and compute the endomorphism ring $F = \mathrm{end}_A(L)$ of L:

```
> Endo1:=MorphismsConstCoeff(S,S,A): Endo1[1]; Endo1[2];
```

$$[\begin{bmatrix} 0 & u^2\,\rho^2 \\ c^2\,(-c^2+u^2) & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}] \quad \begin{bmatrix} d_y & -u\,c^2\,\rho\,d_x + u^3\,d_x\,\rho \\ d_x & u\,\rho\,c^2\,d_y \end{bmatrix}$$

Let us check whether or not we can find idempotents of F defined by means of constant matrices:

```
> Idem1:=IdempotentsConstCoeff(S,Endo1[1],A,0,alpha);
```

$$Idem1 := [[\begin{bmatrix} 1/2 & \frac{u\,\rho\,\alpha_1}{c} \\ \frac{\alpha_1\,c\,(-c^2+u^2)}{u\,\rho} & 1/2 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}],$$

$$[Ore\_algebra, [\text{"}diff\text{"}, \text{"}diff\text{"}], [x,y], [d_x, d_y], [x,y], [u, \rho, c, \alpha_1], 0, [], [-1 - 4\,{\alpha_1}^2\,c^2 + 4\,{\alpha_1}^2\,u^2],$$
$$[x,y], [], [], [diff = [d_x, x], diff = [d_y, y]]]]$$

We obtain the two trivial idempotents 0 and id_F of F and an idempotent of $\mathrm{end}_B(B \otimes_A L)$, where $B = \mathbb{Q}(u,\rho,c)[\alpha_1]/(4\,(u^2-c^2)\,\alpha_1^2 - 1)[d_x, d_y]$, defined by the following matrix:

```
> B:=Idem1[2]: P[2]:=Idem1[1,1]; Q[2]:=Factorize(Mult(S,P[2],B),S,B);
```

$$P_2 := \begin{bmatrix} 1/2 & \frac{u\,\rho\,\alpha_1}{c} \\ \frac{\alpha_1\,c\,(-c^2+u^2)}{u\,\rho} & 1/2 \end{bmatrix} \quad Q_2 := \begin{bmatrix} 1/2 & \frac{\alpha_1\,c^2 - \alpha_1\,u^2}{c} \\ -\alpha_1\,c & 1/2 \end{bmatrix}$$

We can check that the matrices P_2 and Q_2 satisfy $P_2^2 = P_2$ and $Q_2^2 = Q_2$:

```
> VERIF1:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),alpha[1]^4=1/16/(u^2-c^2)^2,
> simplify(evalm(Mult(P[2],P[2],B)-P[2]))));
> VERIF2:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),alpha[1]^4=1/16/(u^2-c^2)^2,
> simplify(evalm(Mult(Q[2],Q[2],B)-Q[2]))));
```

$$VERIF1 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad VERIF2 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

As the matrices P_2 and Q_2 are idempotents of $B^{2\times 2}$, we know that the B-modules $\ker_B(.P_2)$, $\mathrm{im}_B(.P_2) = \ker_B(.(I_2 - P_2))$, $\ker_B(.Q_2)$ and $\mathrm{im}_B(.Q_2) = \ker_B(.(I_2 - Q_2))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of those free B-modules:

```
> X1:=SyzygyModule(P[2],B):X2:=SyzygyModule(evalm(1-P[2]),B):
> X:=stackmatrix(X1,X2);
> Y1:=SyzygyModule(Q[2],B): Y2:=SyzygyModule(evalm(1-Q[2]),B):
> Y:=stackmatrix(Y1,Y2);
```

$$X := \begin{bmatrix} -2\,\alpha_1\,c\,u^2 + 2\,\alpha_1\,c^3 & u\,\rho \\ -2\,\alpha_1\,c^3 + 2\,\alpha_1\,c\,u^2 & u\,\rho \end{bmatrix} \quad Y := \begin{bmatrix} -2\,\alpha_1\,c & -1 \\ -2\,\alpha_1\,c & 1 \end{bmatrix}$$

We can easily check that $X\,P_2\,X^{-1}$ and $Y\,Q_2\,Y^{-1}$ are the block-diagonal matrices $\mathrm{diag}(0,1)$:

```
> VERIF1:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),Mult(X,P[2],
> LeftInverse(X,B),B)));
> VERIF2:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),Mult(Y,Q[2],
> LeftInverse(Y,B),B)));
```

$$\mathit{VERIF1} := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} \quad \mathit{VERIF2} := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

Then, the matrix S is equivalent to the block-diagonal matrix $Y\,S\,X^{-1}$ defined by:

```
> S_dec:=simplify(subs(alpha[1]^2=1/4/(u^2-c^2),Mult(Y,S,LeftInverse(X,B),B)));
```

$$\mathit{S\_dec} := \begin{bmatrix} \frac{c\,d_y + 2\,d_x\,\alpha_1\,c^2 - 2\,d_x\,\alpha_1\,u^2}{2\,\alpha_1(u^2-c^2)} & 0 \\ 0 & \frac{c\,d_y - 2\,d_x\,\alpha_1\,c^2 + 2\,d_x\,\alpha_1\,u^2}{2\,\alpha_1(u^2-c^2)} \end{bmatrix}$$

This last result can be directly obtained as follows:

```
> simplify(subs(alpha[1]^2=1/4/(u^2-c^2),HeuristicDecomposition(S,P[2],B)
> [1]));
```

$$\begin{bmatrix} \frac{c\,d_y + 2\,d_x\,\alpha_1\,c^2 - 2\,d_x\,\alpha_1\,u^2}{2\,\alpha_1(u^2-c^2)} & 0 \\ 0 & \frac{c\,d_y - 2\,d_x\,\alpha_1\,c^2 + 2\,d_x\,\alpha_1\,u^2}{2\,\alpha_1(u^2-c^2)} \end{bmatrix}$$

If we denote by

```
> G:=diag(X,1): H:=diag(Y,1): Z:=Mult(G,U,B); T:=Mult(H,V,B);
```

$$Z := \begin{bmatrix} 0 & -2\,\alpha_1\,c\,u^2 + 2\,\alpha_1\,c^3 & u\,\rho \\ 0 & -2\,\alpha_1\,c^3 + 2\,\alpha_1\,c\,u^2 & u\,\rho \\ u\,\rho & c^2 & 0 \end{bmatrix} \quad T := \begin{bmatrix} -2\,\alpha_1\,c & -1 & 2\,\alpha_1\,c\,u \\ -2\,\alpha_1\,c & 1 & 2\,\alpha_1\,c\,u \\ 1 & 0 & 0 \end{bmatrix}$$

then the matrix R is equivalent to the simple block-diagonal matrix $T\,R\,Z^{-1}$ defined by:

```
> simplify(subs(alpha[1]^2=1/4/(u^2-c^2),simplify(Mult(T,R,LeftInverse(Z,B),
> B))));
```

$$\begin{bmatrix} \frac{c\,d_y + 2\,d_x\,\alpha_1\,c^2 - 2\,d_x\,\alpha_1\,u^2}{2\,\alpha_1(u^2-c^2)} & 0 & 0 \\ 0 & \frac{c\,d_y - 2\,d_x\,\alpha_1\,c^2 + 2\,d_x\,\alpha_1\,u^2}{2\,\alpha_1(u^2-c^2)} & 0 \\ 0 & 0 & d_x \end{bmatrix}$$

If $\mathcal{F}$ denotes an A-module (e.g., $\mathcal{F} = C^{\infty}(\mathbb{R}^2)$), using the relation $2\,\alpha_1\,(c^2 - u^2) = -1/(2\,\alpha_1)$, we then obtain that the linear system $\ker_{\mathcal{F}}(R.)$ is equivalent to the following one

$$\left\{ \begin{array}{l} (d_x - 2\,c\,\alpha_1\,d_y)\,\zeta_1 = 0, \\ (d_x + 2\,c\,\alpha_1\,d_y)\,\zeta_2 = 0, \\ d_x\,\zeta_3 = 0, \end{array} \right. \quad \Leftrightarrow \quad \left\{ \begin{array}{l} \zeta_1 = \phi(y + 2\,c\,\alpha_1\,x), \\ \zeta_2 = \psi(y - 2\,c\,\alpha_1\,x), \\ \zeta_3 = C, \end{array} \right.$$

where ϕ and ψ are two arbitrary functions of $\mathcal{F}$ and C an arbitrary constant.